

[> **restart:**

Ranada2.mws (RMKiehn 9/10/97)

A Propagating solution with Longitudinal B is computed using Maple.. By modifying the Ranada stereographic example slightly it is possible to generate a propagating solution to the Maxwell Faraday equations that is irreducibly 3 dimensional. The Electric field in the example is always transverse to the surface of propagating phase, but the B field is not. The fields are constructed from a 4 vector potential, and by Poincare's lemma satisfy the partial differential equations equivalent to the Maxwell - Faraday equations $\text{curl } E + \text{partial } dB/dt = 0$.

This example has nothing to do with a choice of a metric, a choice of a group, CPT invariance, gauge invariance, or non-abelian Lie algebras, or other arguments utilized by Evans and others to support the existence of an irreducibly 3 dimensional B field. These arguments are not relevant.

However, it is a fact that this constructive solution is a counter example to the statement that a non-transverse B field cannot exist.

In fact there exist many solutions to Maxwell's equations that are irreducibly 3-dimensional fields: the B field has three components almost everywhere. That is, the B field cannot be transverse globally. The issue is one of the topology induced on the base domain by the functional form of the 4 vector potential. When the associated 1-form of Action is not integrable in the sense of Frobenius, then minimum Pfaff dimension or class of the form is 3. These ideas have been known for at least 25 years in association with EM theory. Certainly the ideas (not in EM form) were known by the students of Pfaff's problem, an area of research that dates back more than 100 years. Bateman (1914) had solutions to Maxwell's equations that were of the same equivalence class (but not exactly the same) as the example given below.

[>

> **restart;with(liesymm):with(linalg):with(plots):setup(x,y,z,t);**

Warning, new definition for close

Warning, new definition for norm

Warning, new definition for trace

[x, y, z, t]

> **deform(x=0,y=0,z=0,t=0,a=const,b=const,c=const,k=const,mu=const,omega=const,m=const);**

deform(x = 0, y = 0, z = 0, t = 0, a = const, b = const, c = const, k = const, μ = const, ω = const, m = const)

> **dR:=[d(x),d(y),d(z),d(t)];**

dR := [d(x), d(y), d(z), d(t)]

Define a characteristic denominator, and then specify the four functions that are the covariant components of the Action 1-form.

> **p:=2;e:=0;c:=1;a:=1;b:=1;n:=4;rrxy:=1+a*(x)^p+b*(y)^p+c*(z)^p+e*t^p;ff:=cos(k*z-omega*t)/rrxy^(n/p);**

> ;

p := 2

e := 0

c := 1

a := 1

b := 1

n := 4

$$ff := \frac{\cos(-kz + \omega t)}{(1 + x^2 + y^2 + z^2)^2}$$

A choice for the 4 potentials. Note that p has been chosen as 2 to yield the quadratic form, n is taken to be 4 to fit the Rana format, the phase propagation is presumed to be in the positive z direction, and the speed of propagation is ω/k .

> **A1:=y*ff;A2:=-x*ff;A3:=k*cos(k*z-omega*t);A4:=omega*cos(k*z-omega*t);**

$$A1 := \frac{y \cos(-kz + \omega t)}{(1 + x^2 + y^2 + z^2)^2}$$

$$A2 := -\frac{x \cos(-kz + \omega t)}{(1 + x^2 + y^2 + z^2)^2}$$

$$A3 := k \cos(-kz + \omega t)$$

$$A4 := \omega \cos(-kz + \omega t)$$

> **Action:=A1*d(x)+A2*d(y)+A3*d(z)-A4*d(t);**

$$\text{Action} := \frac{y \%1 d(x)}{(1 + x^2 + y^2 + z^2)^2} - \frac{x \%1 d(y)}{(1 + x^2 + y^2 + z^2)^2} + k \%1 d(z) - \omega \%1 d(t)$$

$$\%1 := \cos(-kz + \omega t)$$

> **F:=wcollect(d(Action));**

$$F := \frac{y \%2 \omega ((d(x)) \&^{\wedge}(d(t)))}{(1 + x^2 + y^2 + z^2)^2} + \left(-\frac{\%1 (1 - 3x^2 + y^2 + z^2)}{(1 + x^2 + y^2 + z^2)^3} - \frac{\%1 (1 + x^2 - 3y^2 + z^2)}{(1 + x^2 + y^2 + z^2)^3} \right) ((d(x)) \&^{\wedge}(d(y)))$$

$$- \frac{x \%2 \omega ((d(y)) \&^{\wedge}(d(t)))}{(1 + x^2 + y^2 + z^2)^2} - \frac{y (\%2 k + \%2 k x^2 + \%2 k y^2 + \%2 k z^2 - 4 \%1 z) ((d(x)) \&^{\wedge}(d(z)))}{(1 + x^2 + y^2 + z^2)^3}$$

$$+ \frac{x (\%2 k + \%2 k x^2 + \%2 k y^2 + \%2 k z^2 - 4 \%1 z) ((d(y)) \&^{\wedge}(d(z)))}{(1 + x^2 + y^2 + z^2)^3}$$

$$\%1 := \cos(-kz + \omega t)$$

$$\%2 := \sin(-kz + \omega t)$$

F is the electromagnetic 2-form in covariant language for all diffeomorphisms. It is gauge invariant with respect to all closed 1-form additions to the 1-form of Action.

The three components of the Vector potential are:

> **A:=[A1,A2,A3];**

$$A := \left[\frac{y \cos(-kz + \omega t)}{(1 + x^2 + y^2 + z^2)^2}, -\frac{x \cos(-kz + \omega t)}{(1 + x^2 + y^2 + z^2)^2}, k \cos(-kz + \omega t) \right]$$

The three components of the Magnetic field are:

> **B:=(curl(A,[x,y,z]));B1:=(B[1]);B2:=(B[2]);B3:=factor(B[3]);**

$$B1 := \frac{x \sin(-kz + \omega t) k}{(1 + x^2 + y^2 + z^2)^2} - 4 \frac{x \cos(-kz + \omega t) z}{(1 + x^2 + y^2 + z^2)^3}$$

$$B2 := \frac{y \sin(-kz + \omega t) k}{(1 + x^2 + y^2 + z^2)^2} - 4 \frac{y \cos(-kz + \omega t) z}{(1 + x^2 + y^2 + z^2)^3}$$

$$B3 := -2 \frac{\cos(-kz + \omega t) (1 - x^2 - y^2 + z^2)}{(1 + x^2 + y^2 + z^2)^3}$$

Note that B3 is not zero (almost everywhere, except on the hyperbola of 1 sheet) and is perpendicular to the propagating phase front.

The three components of the Electric Field are:

```

>
> E:=[-diff(A4,x)-diff(A[1],t),-diff(A4,y)-diff(A[2],t),-diff(A4,z)-diff(A[3],t)]:
E1:=factor(E[1]);E2:=factor(E[2]);E3:=factor(E[3]);CURLE:=curl(E,[x,y,z]);

```

$$E1 := \frac{y \sin(-kz + \omega t) \omega}{(1 + x^2 + y^2 + z^2)^2}$$

$$E2 := -\frac{x \sin(-kz + \omega t) \omega}{(1 + x^2 + y^2 + z^2)^2}$$

$$E3 := 0$$

$$CURLE := \left[-\frac{x k \cos(-kz + \omega t) \omega}{(1 + x^2 + y^2 + z^2)^2} - 4 \frac{x \%1 \omega z}{(1 + x^2 + y^2 + z^2)^3}, -\frac{y k \cos(-kz + \omega t) \omega}{(1 + x^2 + y^2 + z^2)^2} - 4 \frac{y \%1 \omega z}{(1 + x^2 + y^2 + z^2)^3}, \right.$$

$$\left. -2 \frac{\%1 \omega}{(1 + x^2 + y^2 + z^2)^2} + 4 \frac{x^2 \%1 \omega}{(1 + x^2 + y^2 + z^2)^3} + 4 \frac{y^2 \%1 \omega}{(1 + x^2 + y^2 + z^2)^3} \right]$$

$$\%1 := \sin(-kz + \omega t)$$

It would appear that this "wave" is TE (E3 in the direction of advancing phase is zero.) The example demonstrates that all three components of E must be zero is false.

The Topological Parity 4 form (Second Poincare invariant or $2E \cdot B \, dx^i dy^j dz^k dt$) vanishes implying that the 1 form of Action defines a contact manifold. Therefore the integral of the 3-form of topological torsion over a closed domain, not a boundary, is a non-zero topological invariant. The result $E \cdot B = 0$ is crucial for the existence of characteristics. The integral of the Topological 3-form over a 3-D boundary of space time is zero. But the same integral over a 3D closed chain which is not a boundary (a cycle) is not zero, and in certain circumstances is called the Hopf index (or relative integral invariant).

For transverse waves (with the wave vector in the direction of the vector potential) the Topological Torsion 3-form vanishes.

```

> EdotB:=factor(innerprod(E,B));EdotE:=innerprod(E,E);

```

$$EdotB := 0$$

$$EdotE := \frac{\sin(-kz + \omega t)^2 \omega^2 (y^2 + x^2)}{(1 + x^2 + y^2 + z^2)^4}$$

The Torsion current.

```

> ExA:=crossprod(E,A):Bphi:=[B1*A4,B2*A4,B3*A4]:
> TORS:=evalm(ExA+A4*B):
> AdotB:=factor(inner(A,B)):Helicity:=AdotB;

```

$$Helicity := -2 \frac{k \cos(-kz + \omega t)^2 (1 - x^2 - y^2 + z^2)}{(1 + x^2 + y^2 + z^2)^3}$$

```

> TORSION:=[factor(TORS[1]),factor(TORS[2]),factor(TORS[3]),AdotB];

```

$$TORSION := \left[-4 \frac{x \omega \%1^2 z}{(1 + x^2 + y^2 + z^2)^3}, -4 \frac{y \omega \%1^2 z}{(1 + x^2 + y^2 + z^2)^3}, -2 \frac{\%1^2 (1 - x^2 - y^2 + z^2) \omega}{(1 + x^2 + y^2 + z^2)^3}, -2 \frac{k \%1^2 (1 - x^2 - y^2 + z^2)}{(1 + x^2 + y^2 + z^2)^3} \right]$$

$$\%1 := \cos(-kz + \omega t)$$

The 4 vector of Torsion Current (and the fourth component which is the helicity) are proportional to the SQUARE of the oscillatory term phase term.

The 4Divergence of the Torsion current is:

```

> DIV4DofT:=factor(diverge(TORSION,[x,y,z,t]));

```

$$DIV4DofT := 0$$

[establishing the "conservation law" for topological torsion.

[These fields satisfy the Maxwell-Faraday equations for E and B, by construction and the use of Poincare's lemma for C2 functions.

[Now just to prove that the solutions satisfy the Maxwell equation:

```
[ > DIVB:=diverge(B,[x,y,z]);  
                                DIVB := 0  
[ > MAX1:=CURLE[1]+diff(B[1],t);MAX2:=CURLE[2]+diff(B[2],t);MAX3:=CURLE[3]+diff(B[3]  
  ,t);  
                                MAX1 := 0  
                                MAX2 := 0  
                                MAX3 := 0
```

No mention is made of the field excitations, D and H, nor of the charge distributions and constitutive relations that might be compatible with these field intensities, E and B. However, as the Torsion Current is closed, it is a candidate for the charge current distribution of classical electromagnetic theory. This result ($E \cdot B = 0$) establishes the criteria for the existence of characteristics (e.g. waves).

The problem is to find solutions to the partial differential equations, $\text{curl} H - \text{partial} D / \text{dt} = T$ and $\text{div} D = A \cdot B$, where the RHS is given. That is a problem for another day. It is not apparent that J and rho are zero for any realizable constitutive equation with the symmetries of a Lorentz vacuum. That is another problem, that is not addressed herein.

RMK.